HISTORICAL EVENT FILTER WITH DATE AND TIME FILTER APPLIED

Local Tag created to store Date & Time picker. 
	L_SOE_FilterToDate 		[Data Type = STRING]													
	L_SOE_FilterToTime		[Data Type = STRING]	
	L_SOE_FilterFromDate	[Data Type = STRING]													
	L_SOE_FilterFromTime	[Data Type = STRING]	

Now button will set “To Time” to “Now” (both date and time). 
The custom script below will append date & time filter to current filter applied. 
This is done by adding a tie in to SA Include-> AlarmPage.ci-> FUNCTION _AlarmPage_UpdateFilter().
*******************************************************************************************
FUNCTION _AlarmPage_UpdateFilter()

	INT		nAlarmListAn = DspGetAnFromName("AlarmList");
	INT		nError;
	INT		hFilter;

	// Reset the alarm list scroll position to 0
	_ArrayView_SetYOffset(nAlarmListAn, 0);

	// Explicitly include system events if it is SoE and filter is specified
	STRING	sFilterGroup = AlarmPage_GetFilterGroup("FltrGrp");
	INT		bForceSystemEvents = FALSE;
	INT		nType = PageGetInt("__Type");

	IF (nType = 15 OR nType = 16) THEN
		bForceSystemEvents = FilterGroup_GetItemCount(sFilterGroup) > 0;

	END
		
	hFilter = AlarmFilterEditOpen(nAlarmListAn);
	IF hFilter <> BAD_HANDLE THEN
		// Inject opening query clause for explicitly including system events
		IF (bForceSystemEvents) THEN
			nError = AlarmFilterEditSet(hFilter, "(");
		ELSE
			nError = AlarmFilterEditSet(hFilter, "");
		END

		IF nError = 0 THEN
			// Process all filters
			AlarmFilter_AppendFilterGroup(hFilter, sFilterGroup);

			// Inject closing query clause for explicitly including system events
			IF (bForceSystemEvents) THEN
				//AlarmFilterEditAppend(hFilter, ") OR StateNumeric=0");//default
				AlarmFilterEditAppend(hFilter, ")");
			END

			nError = AlarmFilterEditCommit(hFilter);
		END
		
		
		AlarmFilterEditClose(hFilter);
	END
	
	_AlarmPage_UpdateFilterText(_AlarmPage_GetCurrentFilter());
	
	//********************************************************************************************
	//*******VICOSOL TIE-IN TO BE ABLE TO SORT EVENT LIST ON DATE&TIME****************************
	//********************************************************************************************
	//
	VIC_AlarmPage_UpdateSOE();
	//
	//********************************************************************************************
	//********************************************************************************************
	
END

//-----------------------------------------------------------------------------------------------




*******************************************************************************************
Custom script to append date and time filters

/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//							VICOSOL AS
//FUNCTION NAME: 			VIC_SOEFilterInit()
//REVISION: 				01
//DESCRIPTION: 				Historical Event List: This function is used to set initial values when Historical Event screen is opened.
//DATE OF REVISION: 		2021-10-06
//REVISED BY:				Rune Johannessen
//REVISION DESCRIPTION:		INITIAL GENERATION
// 
//ADDITONAL DESCRIPTION:	This function is called from Historical Event screen on page entry. 
//							Page Properties - Events - On Page Entry
//
//							Upon initialization End date and time is set to current time and Start date and time is set to current time.
//							
//							
//							
//							
//							
//
//							SPECIAL NOTE:
//							
//							________________________________________________						
//							©Vicosol AS All rights reserved 
/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////

FUNCTION VIC_SOEFilterInit()
	INT TimeSpan;
	STRING FromDate;
	STRING FromTime;
	STRING ToDate;
	STRING ToTime;
	
	//Initial timespan is set here. Enter time span i seconds from end date and time. 
	//Example:
	//Setting Timespan to 86400 (24hr) will set the start date to "yesterday" with same time as current. 
	//
	TimeSpan = 86400;
	//
	//Upon initialization End date and time is set to current time
	ToDate = Date();													
	ToTime =Time();
	//Upon initialization Start date and time is set to current time - TimeSpan	
	FromDate = TimeToStr(DateSub(StrToDate(ToDate),TimeSpan),2);
	FromTime = Time();
	
	//Function output values
	L_SOE_FilterToDate = ToDate;													
	L_SOE_FilterToTime = ToTime;	
	L_SOE_FilterFromDate = FromDate;													
	L_SOE_FilterFromTime = FromTime;	
	

END
//-----------------------------------------------------------------------------------------------


/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//							VICOSOL AS
//FUNCTION NAME: 			VIC_SOETime_Now()
//REVISION: 				01
//DESCRIPTION: 				Historical Event List: This function will set "To Time" to Now.
//DATE OF REVISION: 		2021-10-06
//REVISED BY:				Rune Johannessen
//REVISION DESCRIPTION:		INITIAL GENERATION
// 
//ADDITONAL DESCRIPTION:	This function is called from button on SOE page
//							
//
//													
//							
//							
//							
//							
//
//							SPECIAL NOTE:
//							
//							________________________________________________						
//							©Vicosol AS All rights reserved 
/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////

FUNCTION VIC_SOETime_Now()

	STRING ToDate;
	STRING ToTime;
	
	ToDate = Date();													
	ToTime =Time();

//Function output values
	L_SOE_FilterToDate = ToDate;													
	L_SOE_FilterToTime = ToTime;	
END
//-----------------------------------------------------------------------------------------------


/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
//							VICOSOL AS
//FUNCTION NAME: 			VIC_AlarmPage_UpdateSOE()
//REVISION: 				01
//DESCRIPTION: 				Historical Event List: This function will take current SOE filter and append Date&Time filter.
//DATE OF REVISION: 		2021-10-06
//REVISED BY:				Rune Johannessen
//REVISION DESCRIPTION:		INITIAL GENERATION
// 
//ADDITONAL DESCRIPTION:	This function is called from SA Include-> AlarmPage.ci-> FUNCTION _AlarmPage_UpdateFilter()
//							
//
//													
//							
//							
//							
//							
//
//							SPECIAL NOTE:
//							The tie-in to SA Include must be manually updated if new version AVEVA Plant Scada is installed. 
//							________________________________________________						
//							©Vicosol AS All rights reserved 
/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
	
FUNCTION VIC_AlarmPage_UpdateSOE()
	
	INT nSOEAn = DspGetAnFromName("AlarmList");
	IF (nSOEAn < 1) THEN
		RETURN;
	END

    INT     hFilter;
    INT     nError = 0;
	
	INT FromTime; 
	INT ToTime;
	
	
	FromTime = DateAdd(StrToDate(L_SOE_FilterFromDate),StrToTime(L_SOE_FilterFromTime)); 	//Converts Start Date and Time to Integer value
	ToTime = DateAdd(StrToDate(L_SOE_FilterToDate),StrToTime(L_SOE_FilterToTime));			//Converts End Date and Time to Integer value

	hFilter = AlarmFilterEditOpen(nSOEAn);
	IF hFilter <> BAD_HANDLE THEN

    	IF nError = 0 THEN
    
    		AlarmFilterEditAppend(hFilter,";");
			AlarmFilterEditAppend(hFilter, "TAG=*" + ";TIME>" + IntToStr(FromTime) +";TIME<" + IntToStr(ToTime)); // The ";" between <filter_expr> is interpreted as AND 
            nError = AlarmFilterEditCommit(hFilter);

		END
		AlarmFilterEditClose(hFilter);
	END
END

//-----------------------------------------------------------------------------------------------

